Timer0 :





Das nächste Programm sorgt dafür, daß die LED's an Port B im Abstand von ca. einer Sekunde 


ein- und ausgeschaltet werden. Dazu wird ein von Timer0 ausgelöster Interrupt genutzt.  





Zum Einsatz kommen folgende Register: SP, DDRB, TIMSK, TCNT0, TCCR0.





Der Timer wird eingeschaltet über Register TCCR0. Solange dort 000 in den untersten Bits 


ist, tut sich garnichts (da der Timer ausgeschaltet ist). Wenn 101 in die untersten 


Bits geschrieben wird, dann wird Timer0 gestartet mit einem Vorteiler von 1024. Das bedeutet, 


daß Timer0 immer nach 1024 Quartz-Takten einen Zähler inkrementiert, der in dem Register 


TCNT0 gehalten wird. Sobald dieses Register überfließt (also durch ein weiteres 


Inkrementieren wieder auf 0000 0000 springt), wird der Interrupt ausgelöst. 





Damit dieser Interrupt tatsächlich einen Interrupt-Handler startet, muß der Timer0 


Interrupt explizit aktiviert werden (über Register TIMSK) und außerdem sind mit dem 


Befehl "sei" generell Interrupts zu aktivieren. 





Das Zählregister TCNT0 ist dabei vorbelegt mit dem Wert 256 - 244. Timer0 inkrementiert 


also 244 mal, bevor er den Interrupt auslöst. Damit sind seit dem (Neu-)Starten des Timer 


244 * 1024 Taktzyklen vergangen, also ca. 250 000 Taktzyklen. Bei einem 8MHz Quartz 


sind das ca. 1/32 Sekunden. Damit die LED's im Abstand von ca. 1 Sekunde blinken, wird 


im Interrupt-Handler ein weiterer Zähler geführt. Der sorgt dafür, daß nur nach jeweils 


32 Interrupts die Bits umgedreht werden, die an PORTB die Lämpchen leuchten lassen.





Hier ein Programmbeispiel:





timer0.asm





.include "4433def.inc"  ;ggf. anpassen.


.def temp = r16


.def counter = r17


.def leds = r18


.equ start = 256 - 244 ;  bei 8 MHz entspricht das ca. 1/32 sec


.equ loops = 32;





rjmp reset           ; Reset Handler


reti					       ; IRQ0 Handler


reti					       ; IRQ1 Handler


reti                 ; Timer1 Capture Handler


reti                 ; Timer1 compare Handler


reti                 ; Timer1 Overflow Handler


rjmp timer0          ; Timer0 Overflow Handler


reti                 ; SPI Transfer Complete Handler


reti                 ; UART RX Complete Handler : RXCIE


reti                 ; UDR Empty Handler


reti                 ; UART TX Complete Handler


reti                 ; ADC Conversion Complete Interrupt Handler


reti                 ; EEPROM Ready Handler


reti                 ; Analog Comparator Handler





reset:                


	ldi temp, RAMEND   


	out SP, temp	; set stack pointer





	ldi temp, 0xFF


	out DDRB, temp ; PORTB configured for output


	ldi temp, 0xFF


	out PORTB, leds ; LEDs ausschalten





	ldi temp, 1<<TOIE0  ; 0000 0010  


	out TIMSK, temp  ; timer 0 interrupt ein


	ldi temp, start 


	out TCNT0, temp ; Startwert des Timers


	ldi temp, 0b00000101 


	out TCCR0, temp  ; ; Timer starten mit Vorteiler 1024





	ldi counter, loops


	ldi leds, 0xFF





	sei ; interrupts generell aktivieren





main:


	loop:


		rjmp loop 








;--------------------------------------------------


timer0:


	dec counter		


	brne restart


		ldi counter, loops	


		com leds


		out PORTB, leds


	restart:


		ldi temp, start 


		out TCNT0, temp ; Timer fängt wieder von vorn an


reti











_____________________________________________________________________





Timer1 :





Das folgende Programm hat den gleichen Effekt, aber diesmal wird Timer1 genutzt, 


dessen Zähler mit 16 Bit arbeitet und der damit bis zu 65536 mal inkrementieren kann bis 


zum Auslösen des Interrupts. Konkret werden die beiden Zähl-Register TCNT1H und TCNT1L 


vorbelegt mit insgesamt 65536 - 7812. Damit wird der Interrupt ausgelöst nach 


7812 * 1024 Taktzyklen, also 7999488, was bei einem 8 MHz Quartz annähernd 1 Sekunde dauert. 





Der Vollständigkeit halber sei angemerkt, daß in beiden Fällen (timer0.asm und timer1.asm) 


die eigentliche Rechenleistung des MC weitgehend ungenutzt bleibt. Das Hauptprogramm macht 


nichts, außer sich selbst mit einer Schleife zu beschäftigen. Der MC ist also für weitere 


Aufgaben durchaus noch einsetzbar (Taster abfragen, Motoren steuern etc.).   





   main:


	loop:


		rjmp loop 





Hier das Beispielprogramm:





timer1.asm





.include "4433def.inc"


.def temp = r16


.def leds = r17


.equ start = 65536 - 7812;   2 hoch 16 minus 7812


.equ highValue = high(start)


.equ lowValue  = low(start) 





rjmp reset           ; Reset Handler


reti					       ; IRQ0 Handler


reti					       ;IRQ1 Handler


reti                 ; Timer1 Capture Handler


reti                 ; Timer1 compare Handler


rjmp timer1          ; Timer1 Overflow Handler


reti                 ; Timer0 Overflow Handler


reti                 ; SPI Transfer Complete Handler


reti                 ; UART RX Complete Handler : RXCIE


reti                 ; UDR Empty Handler


reti                 ; UART TX Complete Handler


reti                 ; ADC Conversion Complete Interrupt Handler


reti                 ; EEPROM Ready Handler


reti                 ; Analog Comparator Handler





reset:                


	ldi temp, RAMEND   


	out SP, temp	; set stack pointer





	ldi temp, 0xFF


	out DDRB, temp ; PORTB configured for output


	ldi leds, 0xFF


	out PORTB, leds ; LEDs ausschalten





	ldi temp, 1<<TOIE1  ; 1000 0000  


	out TIMSK, temp  ; Timer 1 interrupt ein





	; Startwert des Timers einstellen:


	ldi temp, highValue 


	out TCNT1H, temp  


	ldi temp, lowValue 


	out TCNT1L, temp  





	ldi temp, 0b00000101 


	out TCCR1B, temp  ; Timer 1 starten mit Vorteiler 1024





	sei ; interrupts generell aktivieren





main:


	loop:


		rjmp loop 








;--------------------------------------------------


timer1:


	com leds


	out PORTB, leds





	; Startwert des Timers neu einstellen:


	ldi temp, highValue 


	out TCNT1H, temp  


	ldi temp, lowValue 


	out TCNT1L, temp  





reti








__________________________________________________________________________________


